# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=sha1collisiondetection
pkgver=106.a06f1ad
pkgrel=2
pkgdesc='Library and command line tool to detect SHA-1 collision in a file'
arch=('x86_64' 'aarch64')
groups=('blackarch' 'blackarch-crypto')
url='https://github.com/cr-marcstevens/sha1collisiondetection'
license=('MIT')
makedepends=('git')
source=("git+https://github.com/cr-marcstevens/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

package() {
  cd $pkgname

  PREFIX="$pkgdir/usr" make install

  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md
  install -Dm 644 LICENSE.txt "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

